*******		REPLICATION FILES
*******		Climate Variability and Irregular Migration to the European Union 
*******		Global Environmental Change 
*******		Fabien Cottier and Idean Salehyan
*******		Replication: Sensitivity analysis (S2) migration rate variable (Tables A.9-11, Figures A.8-9)
*******		This version: April 2021

* note: stata packages required for running models
* - coefplot
* - crossfold
* - plottig


clear all

set more off
set scheme plottig
cd "path/to/replication/directory"

* load data
use "data_quarter.dta", clear 

* load stata risk scripts
qui do scripts/script_rr_spei_082019

* duplicates data
duplicates list cowc year quarter
duplicates tag cowc year quarter, gen(_dupl)
duplicates drop cowc year quarter, force

* set time series
sort cowc year quarter
gen quarter_int = real(regexs(1)) if regexm(quarter,"([0-9]+)")
gen quarterS=(year-2005)*4+quarter_int
order cowc nationality year quarter quarter_int quarterS
tsset cowc quarterS
 
* gen additional variables
encode continent, gen(contN)
recode contN (5=.)


* generate rate migration variable
gen Rmigrq_exbalk=(nmigrq/pop)*(10^5)
gen Rmigrq_exbalk_ln=log(Rmigrq_exbalk+1)

* generate lag quartely migration variables
by cowc: gen nmigrq_exbalk_ln_1tl = nmigrq_exbalk_ln[_n-1]
by cowc: gen nmigrq_exbalk_ln_2tl = nmigrq_exbalk_ln[_n-2]
by cowc: gen nmigrq_exbalk_ln_3tl = nmigrq_exbalk_ln[_n-3]
by cowc: gen nmigrq_exbalk_ln_4tl = nmigrq_exbalk_ln[_n-4]


by cowc: gen Rmigrq_exbalk_1tl = Rmigrq_exbalk[_n-1]
by cowc: gen Rmigrq_exbalk_2tl = Rmigrq_exbalk[_n-2]
by cowc: gen Rmigrq_exbalk_3tl = Rmigrq_exbalk[_n-3]
by cowc: gen Rmigrq_exbalk_4tl = Rmigrq_exbalk[_n-4]
by cowc: gen Rmigrq_exbalk_ln_1tl = Rmigrq_exbalk_ln[_n-1]
by cowc: gen Rmigrq_exbalk_ln_2tl = Rmigrq_exbalk_ln[_n-2]
by cowc: gen Rmigrq_exbalk_ln_3tl = Rmigrq_exbalk_ln[_n-3]
by cowc: gen Rmigrq_exbalk_ln_4tl = Rmigrq_exbalk_ln[_n-4]


* gen dummies variables for sample splitting: low agriculture / high agriculture
qui xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* i.year if migr100_exbalk==1, ///
 fe vce(cluster cowc)

sum agriLab if e(sample) & year==2010, d
scalar agriMed=r(p50)
gen agri_H=0 if e(sample) & year==2010
recode agri_H (0=1) if agriLab-agriMed > 0 & year==2010
by cowc: replace agri_H = agri_H[21] if missing(agri_H)

* generate dummies variable for extreme weather (cut-off 10 percentile / 90 percentile)
qui xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* wmean_speiy /// 
 i.year i.quarter_int if migr100_exbalk==1, ///
 fe vce(cluster cowc)

centile (wmean_speiy) if e(sample), centile (10 90)
gen spei_drought = 0 if wmean_speiy!=.
recode spei_drought (0=1) if wmean_speiy <= r(c_1) 
centile (wmean_speiy) if e(sample), centile (10 90)
gen spei_hrain = 0 if wmean_speiy!=.
recode spei_hrain (0=1) if wmean_speiy >=  r(c_2) 


*******		  Macros

global SPEI_Y0 wmean_speiy

global SPEI_SQ_Y0 c.wmean_speiy##c.wmean_speiy
 
global SPEI_Y0Y2 L(0 4 8).wmean_speiy

global SPEI_SQ_Y0Y2 c.wmean_speiy##c.wmean_speiy cL4.wmean_speiy##cL4.wmean_speiy /// 
  cL8.wmean_speiy##cL8.wmean_speiy


*******		 Table A.9

* NULL Model // No SPEI term
qui xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_Y0 /// 
	i.year i.quarter_int if migr100_exbalk==1, ///
	fe vce(cluster cowc)
qui xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_*  /// 
	i.year i.quarter_int if e(sample), ///
	fe vce(cluster cowc)
est sto tabA9_m0
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse null model     : " CVrmse
* list countries included in sample 
est resto tabA9_m0
unique cowc if e(sample)
tabulate nationality contN if e(sample)


* Model 1
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_Y0 /// 
	i.year i.quarter_int if migr100_exbalk==1, ///
	fe vce(cluster cowc)
est sto tabA9_m1
* F test spei parameters & aic
test wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse


* Model 2
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_SQ_Y0 ///
 i.year i.quarter_int if migr100_exbalk==1, ///
 fe vce(cluster cowc)
est sto tabA9_m2
* F test spei parameters & aic
test wmean_speiy c.wmean_speiy#c.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse


* Model 3
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_Y0Y2 /// 
	i.year i.quarter_int if migr100_exbalk==1, ///
	fe vce(cluster cowc)
est sto tabA9_m3
* F test spei parameters & aic
test L0.wmean_speiy L4.wmean_speiy L8.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse


* Model 4
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_SQ_Y0Y2 ///
	i.year i.quarter_int if migr100_exbalk==1, ///
	fe vce(cluster cowc)
est sto tabA9_m4
* F test spei parameters & aic
test L0.wmean_speiy L4.wmean_speiy L8.wmean_speiy ///
 cL0.wmean_speiy#cL0.wmean_speiy cL4.wmean_speiy#cL4.wmean_speiy cL8.wmean_speiy#cL8.wmean_speiy
estat ic 
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse




*******		 Table A.10


* NULL Model // agrarian countries
qui xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_Y0 ///
	i.year i.quarter_int if migr100_exbalk==1 & agri_H==1, ///
	fe vce(cluster cowc)
qui xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* ///
	i.year i.quarter_int if e(sample), ///
	fe vce(cluster cowc)
est sto tabA10_m0H
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse null model     : " CVrmse
* list of countries
est resto tabA10_m0H
unique cowc if e(sample)
tabulate nationality contN if e(sample)

* NULL Model // non-agrarian countries
qui xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_Y0 ///
	i.year i.quarter_int if migr100_exbalk==1 & agri_H==0, ///
	fe vce(cluster cowc)
qui xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* ///
	i.year i.quarter_int if e(sample), ///
	fe vce(cluster cowc)
est sto tabA10_m0L
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse null model     : " CVrmse
global CVrmsenull_LA=round(CVrmse,0.001)
* list of countries
est resto tabA10_m0L
unique cowc if e(sample)
tabulate nationality contN if e(sample)


* Model 5 // agrarian countries
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_Y0 ///
  i.year i.quarter_int if migr100_exbalk==1 & agri_H==1, ///
  fe vce(cluster cowc)
est sto tabA10_m5
* F test spei parameters & aic
test wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 6 // non-agrarian countries
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_Y0 ///
  i.year i.quarter_int  if migr100_exbalk==1 & agri_H==0, ///
  fe vce(cluster cowc)
est sto tabA10_m6
* F test spei parameters & aic
test wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse


* Model 7 // agrarian countries
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_SQ_Y0 ///
  i.year i.quarter_int  if migr100_exbalk==1 & agri_H==1, ///
  fe vce(cluster cowc)
est sto tabA10_m7
* F test spei parameters & aic
test wmean_speiy c.wmean_speiy#c.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 8 // non-agrarian countries
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_SQ_Y0 ///
  i.year i.quarter_int  if migr100_exbalk==1 & agri_H==0, ///
  fe vce(cluster cowc)
est sto tabA10_m8
* F test spei parameters & aic
test wmean_speiy c.wmean_speiy#c.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse


* Model 9 // agrarian countries
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_Y0Y2 ///
  i.year i.quarter_int  if migr100_exbalk==1 & agri_H==1, ///
  fe vce(cluster cowc)
est sto tabA10_m9
unique cowc if e(sample)
* F test spei parameters & aic
test L0.wmean_speiy L4.wmean_speiy L8.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 10 // non-agrarian countries
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_Y0Y2 ///
  i.year i.quarter_int  if migr100_exbalk==1 & agri_H==0, ///
  fe vce(cluster cowc)
est sto tabA10_m10
unique cowc if e(sample)
* F test spei parameters & aic
test L0.wmean_speiy L4.wmean_speiy L8.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse


* Model 11 // agrarian countriese
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_SQ_Y0Y2 ///
  i.year i.quarter_int  if migr100_exbalk==1 & agri_H==1, ///
  fe vce(cluster cowc)
est sto tabA10_m11
* F test spei parameters & aic
test L0.wmean_speiy L4.wmean_speiy L8.wmean_speiy ///
 cL0.wmean_speiy#cL0.wmean_speiy cL4.wmean_speiy#cL4.wmean_speiy cL8.wmean_speiy#cL8.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 12 // non-agrarian countries
* low reliance on agriculture
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* $SPEI_SQ_Y0Y2 ///
  i.year i.quarter_int  if migr100_exbalk==1 & agri_H==0, ///
  fe vce(cluster cowc)
est sto tabA10_m12
* F test spei parameters & aic
test L0.wmean_speiy L4.wmean_speiy L8.wmean_speiy ///
 cL0.wmean_speiy#cL0.wmean_speiy cL4.wmean_speiy#cL4.wmean_speiy cL8.wmean_speiy#cL8.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse



*******		 Table A.11



* Model 13 // agrarian countries
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* spei_drought spei_hrain ///
	  i.year i.quarter_int if migr100_exbalk==1 & agri_H==1, fe vce(cluster cowc)
est sto tabA11_m13
* F test spei parameters & aic
test spei_drought spei_hrain
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 14 // non-agrarian countries
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* spei_drought spei_hrain  ///
  i.year i.quarter_int if migr100_exbalk==1 & agri_H==0, fe vce(cluster cowc)
est sto tabA11_m14
* F test spei parameters & aic
test spei_drought spei_hrain
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse


* Model 15 // agrarian countries
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* L(0 4 8).spei_drought L(0 4 8).spei_hrain ///
	  i.year i.quarter_int if migr100_exbalk==1 & agri_H==1, fe vce(cluster cowc)
est sto tabA11_m15
* F test spei parameters & aic
test L0.spei_drought L4.spei_drought L8.spei_drought L0.spei_hrain L4.spei_hrain L8.spei_hrain
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 16 // non-agrarian countries
xtreg Rmigrq_exbalk_ln Rmigrq_exbalk_ln_* L(0 4 8).spei_drought L(0 4 8).spei_hrain  ///
  i.year i.quarter_int if migr100_exbalk==1 & agri_H==0, fe vce(cluster cowc)
est sto tabA11_m16
* F test spei parameters & aic
test L0.spei_drought L4.spei_drought L8.spei_drought L0.spei_hrain L4.spei_hrain L8.spei_hrain
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse



*******		 Relative risk figures




* Figure A.8 (Model 2, Table A.9)

qui rr_est tabA9_m2 "quadratic"


* Figure A.9 (Model 7 & 8, Table A.10)

qui rr_est_Sp tabA10_m7 tabA10_m8 "quadratic"

